

# Constructing ROC Curves with the SAS<sup>®</sup> System

Andrew G. Stead, Organon Teknika Corporation, Durham, NC  
 Karen G. MacDonald, Organon Teknika Corporation, Durham, NC

## ABSTRACT

Medical diagnostics firms manufacture kits to screen individuals for a variety of medical conditions. These kits are designed to detect such conditions based on a test sample from an individual being screened. A particular condition is detected as an abnormal optical density above or below a cutoff value. Before such a kit can be sold, however, the kit obviously must demonstrate good discriminating power between positive and negative samples. It is also necessary to determine where to set the cutoff. Receiver operating characteristic (ROC) curves are often used for this purpose. A method for using SAS<sup>®</sup> to expedite this task is presented.

## INTRODUCTION

Medical diagnostics firms develop screening kits to detect various medical conditions in test specimens. These specimens may arise from one of several body fluids. An individual's test sample is treated with reagents optimized to detect a particular condition. During the test process, the kit's reagents are added to the specimen in a translucent container. After allowing a fixed period of time for reaction of the reagents with the sample, light passing through the container is monitored with a light sensitive device. If the test is sensitive to the condition, a positive specimen will change the characteristics of light passing through the container. If the test is specific, a negative specimen will cause little or no perturbation in the light characteristics of the sample.

Under ideal conditions, the optical density associated with the weakest positive sample will be clearly differentiated from the optical density of the noisiest negative sample, and definition of a cutoff optical density is a simple matter. However, for a variety of reasons, including variability in the severity of the condition from specimen to specimen, variability between specimens and within a specimen, and inherent variability in the optical density signal, the distributions of optical densities of positive and negative samples can overlap in practice. During the development process, the manufacturer must determine how much these distributions overlap. If they widely overlap, the test must be redesigned or scrapped. If they barely overlap, the manufacturer must decide where to place this cutoff in order to minimize false positive and false negative declarations with the test system.

Therefore, before the kit is released for sale, the manufacturer needs objective methodology to demonstrate that the test has good sensitivity and specificity, to determine where to set a cutoff optical density for the kit, and possibly to compare the new kit with other kits. Receiver operating characteristic curves are often used for these purposes.

## THE ROC CURVE

To construct an ROC curve, a fixed number of known negative specimens ( $n$ ) and known positive specimens ( $p$ ) are sampled and prepared. These specimens are then tested in a random sequence with the new kit. Optical densities (OD's) are recorded for each specimen. After sorting by expected reactivity, the resultant data set would resemble that shown in the following table.

**Table 1. ROC Curve Data Set**

Specimen #	Expected Reactivity	OD
1	N	$y_1$
2	N	$y_2$
3	N	$y_3$
.	.	.
.	.	.
$n$	N	$y_n$
$n+1$	P	$y_{n+1}$
$n+2$	P	$y_{n+2}$
$n+3$	P	$y_{n+3}$
.	.	.
.	.	.
$n+p$	P	$y_{n+p}$

A series of cutoffs  $\{X_i, (i=1,2,\dots,c)\}$  is then proposed. For each  $X_i$ , frequencies  $a_i$ ,  $b_i$ ,  $c_i$ , and  $d_i$  as shown in Table 2 are calculated.

**Table 2. Cross Classification Of Results For Cutoff,  $X_i$**

		State of Nature		
		Pos	Neg	Totals
Test Result	Reactive	$a_i$	$b_i$	$a_i + b_i$
	Nonreactive	$c_i$	$d_i$	$c_i + d_i$
	Totals	$a_i + c_i$	$b_i + d_i$	$N_i$

For each such table, estimates of rates of false positivity,

$$FP_1 = 100 (b_i / (b_i + d_i)) ,$$

and true positivity,

$$TP_1 = 100 (a_i / (a_i + c_i)) ,$$


are calculated. A data set resembling the one shown in the Table 3 is then compiled.

**Table 3. Data Set Needed To Construct ROC Curve**

Cutoff	True Positive Rate (TP)	False Positive Rate (FP)
1	TP <sub>1</sub>	FP <sub>1</sub>
2	TP <sub>2</sub>	FP <sub>2</sub>
3	TP <sub>3</sub>	FP <sub>3</sub>
.	.	.
.	.	.
c	TP <sub>c</sub>	FP <sub>c</sub>

Finally, an ROC curve is created by plotting TP versus FP for each cutoff. This results in the following figure.

**Figure 1. ROC Curve**



The area under this curve formed by the locus of points shows the ability of the test to discriminate between positive and negative samples. If the locus of points for different cutoffs runs diagonally from lower left (0,0) to upper right (100,100), then the test is a poor screen as the simultaneous minimum of false positive and false negative rates is only 50%.

If the test provides reliable discriminatory power, this curve will also aid the decision as to where to locate the cutoff. The ultimate choice of the cutoff depends on the nature of the test. For some screening tests, a false negative result may have graver consequences than a false positive, while for others the opposite may be true.

**USING SAS TO CONSTRUCT AN ROC CURVE**

While the steps above are conceptually simple, the process of getting from the data set in Table 1 to the data set in Table 3 is less than straightforward with the SAS system, especially without the SAS/IML® module. The approach discussed here allows the user to accomplish this task using only DATA steps, PROC SORT, PROC FREQ, and PROC GPLOT.

**Method**

With the data set described by Table 1 as input, a new data set is created. Within the data step, a DO LOOP is set up to evaluate each optical density measurement with respect to each prospective cutoff. Within the loop, a qualitative result is assigned to a test outcome variable depending on whether or not the optical density reading is greater than or equal to the prospective cutoff. Also within the loop, the optical density, cutoff, and qualitative result are output to a new data set. The output data set is then sorted by cutoff. PROC FREQ is called with a BY statement and true positive and false positive rates are output by prospective cutoff value. Separate data sets for true and false positive rates for each cutoff are created and then merged by cutoff. Finally, PROC GPLOT is called to plot the true positive rate versus the false positive rate using a different symbol for each cutoff. A macro to accomplish these steps is shown in Appendix A.

**EXAMPLE**

Appendix B shows a hypothetical data set comprised of 100 observations that reflect optical density test results obtained for 50 positive (P) and 50 negative (N) specimens. After running the macro described in Appendix A, the following table shows true and false positive rates obtained with each prospective cutoff.

**Table 4. True and False Positive Rates - Sample**

CUTOFF	TP_RATE	FP_RATE
0.50	100	100
0.55	100	92
0.60	94	86
0.65	94	78
0.70	92	72
0.75	84	68
0.80	76	62
0.85	66	60
0.90	58	60
0.95	52	52
1.00	48	50
1.05	42	42
1.10	42	38
1.15	36	34
1.20	24	34
1.25	16	28
1.30	14	22
1.35	12	20
1.40	8	16
1.45	4	8
1.50	4	0

## APPENDIX B

### Sample Data Set

OBS	State	OD	OBS	State	OD
1	P	0.745	51	N	0.504
2	N	0.790	52	N	1.402
3	P	1.440	53	N	1.094
4	P	0.790	54	N	0.773
5	N	1.477	55	P	0.978
6	P	1.178	56	N	1.246
7	P	0.822	57	P	1.188
8	P	0.837	58	N	1.003
9	N	0.642	59	N	0.934
10	P	1.415	60	P	1.005
11	P	1.168	61	N	1.239
12	P	1.296	62	N	0.812
13	N	0.535	63	P	1.112
14	P	1.138	64	N	1.267
15	P	1.203	65	N	1.477
16	P	1.391	66	P	0.597
17	P	1.366	67	N	0.980
18	N	1.456	68	P	1.225
19	N	1.274	69	N	1.065
20	N	1.360	70	N	1.475
21	P	0.961	71	P	0.770
22	P	0.920	72	N	0.681
23	N	0.678	73	P	0.673
24	N	1.016	74	N	1.410
25	P	0.593	75	P	0.823
26	P	1.163	76	N	0.503
27	N	0.920	77	N	0.931
28	N	0.650	78	P	0.898
29	P	0.709	79	P	0.715
30	P	0.950	80	P	1.125
31	N	0.625	81	N	1.403
32	N	1.341	82	N	1.113
33	P	1.028	83	N	0.725
34	N	0.537	84	N	0.948
35	N	0.714	85	N	1.143
36	P	0.738	86	N	0.659
37	P	0.565	87	N	1.420
38	P	1.185	88	P	0.852
39	P	0.839	89	P	0.861
40	P	0.888	90	N	0.644
41	N	0.574	91	P	1.247
42	N	1.022	92	P	0.792
43	N	0.583	93	P	0.793
44	N	1.359	94	N	1.296
45	N	1.221	95	P	1.515
46	P	1.332	96	P	1.175
47	P	1.238	97	P	0.911
48	N	1.028	98	P	1.012
49	N	0.758	99	P	0.815
50	N	0.557	100	P	1.532


### Contact Author:

Andrew G. Stead  
Organon Teknika Corporation  
100 AKZO Avenue  
Durham, NC 27712  
Tel: 919-620-2390  
Email: STEADA@worldnet.att.net

## APPENDIX A

The following figure shows the resultant ROC curve.

Figure 2. ROC Curve - Sample Data



It is easy to see that the test described by the hypothetical sample data is a poor discriminator of positive and negative specimens.

## CONCLUSIONS

In summary, an ROC curve provides an objective measure of the discriminatory power of a screening test and also an idea of where to place the cutoff. With a modification to include extra columns for OD's from other tests, the SAS code presented here can readily provide a visual method to compare the performance of two or more screening tests.

(SAS® and SAS/IML® are registered trademarks of the SAS Institute Inc. in the USA. ® indicates USA registration.)

## REFERENCES

Swets, JA. Measuring the accuracy of diagnostic systems, Science 240: 1285-1293, June, 1988.

## Macro

```

*****
/* MACRO PURPOSE:  CREATE AN ROC DATASET AND PLOT          */
/*
/* VARIABLES          INTERPRETATION                      */
/*
/* DATAIN           INPUT SAS DATA SET                  */
/* LOWLIM            MACRO VARIABLE LOWER LIMIT FOR CUTOFF */
/* UPLIM             MACRO VARIABLE UPPER LIMIT FOR CUTOFF */
/* NINC              MACRO VARIABLE NUMBER OF INCREMENTS  */
/* I                 LOOP INDEX                           */
/* OD                OPTICAL DENSITY                      */
/* CUTOFF            CUTOFF FOR TEST                      */
/* STATE             STATE OF NATURE                      */
/* TEST              QUALITATIVE RESULT WITH CUTOFF       */
/*
/* DATE WRITTEN     BY                                    */
/*
/* 09-25-96        A. STEAD                               */
*****

```

```

%MACRO ROC(DATAIN,LOWLIM,UPLIM,NINC=20);
OPTIONS MTRACE MPRINT;
DATA ROC;
SET &DATAIN;
LOWLIM = &LOWLIM; UPLIM = &UPLIM; NINC = &NINC;
DO I = 1 TO NINC+1;
  CUTOFF = LOWLIM + (I-1)*((UPLIM-LOWLIM)/NINC);
  IF OD > CUTOFF THEN TEST="R"; ELSE TEST="N";
  OUTPUT;
END;
DROP I;
RUN;

```

```

PROC PRINT;
RUN;

```

```

PROC SORT; BY CUTOFF;
RUN;

```

```

PROC FREQ; BY CUTOFF;
TABLE TEST*STATE / OUT=PCTS1 OUTPCT NOPRINT;
RUN;

```

```

DATA TRUEPOS; SET PCTS1; IF STATE="P" AND TEST="R";
TP_RATE = PCT_COL; DROP PCT_COL;
RUN;

```

```

DATA FALSEPOS; SET PCTS1; IF STATE="N" AND TEST="R";
FP_RATE = PCT_COL; DROP PCT_COL;
RUN;

```

```

DATA ROC; MERGE TRUEPOS FALSEPOS; BY CUTOFF;
IF TP_RATE = . THEN TP_RATE=0.0;
IF FP_RATE = . THEN FP_RATE=0.0;
RUN;

```

```

PROC PRINT;
RUN;

```

```

PROC GGPLOT DATA=ROC;
PLOT TP_RATE*FP_RATE=CUTOFF;
RUN;
%MEND;

```